
Abstract: The POPLmark challenge is a collective experiment intended to assess the
usability of theorem provers and proof assistants in the context of fundamental research on
programming languages. In this report, we present a solution to the challenge, developed
with the Coq proof assistant, and using the �locally nameless� presentation of terms with
binders introduced by McKinna, Pollack, Gordon, and McBride.

Une solution sans noms locaux à l'expérience POPLmark

Résumé : L'expérience collective POPLmark vise à évaluer l'utilisabilité des démonstra-
teurs automatiques et des assistants de preuves dans le contexte de la recherche fondamentale
sur les langages de programmation. Dans ce rapport, nous présentons une solution à cette
expérience, développée à l'aide de l'assistant de preuves Coq, et reposant sur la présenta-
tion �sans noms locaux� des termes avec lieurs introduite par McKinna, Pollack, Gordon, et
McBride.

Mots-clés : POPLmark, Coq, sans noms locaux, alpha-conversion, lieurs, systèmes de
types, métathéorie, système F-sub

Chapter 1

Introduction

1.1 The POPLmark challenge

The POPLmark challenge [ABF+05] is a collective experiment intended to assess the us-
ability of theorem provers and proof assistants in the context of fundamental research on
programming languages. The need for computer assistance when formalizing and proving
properties of programming languages (formal semantics and type systems) is well expressed
in the statement of the challenge:

Many proofs about programming languages are long, straightforward, and te-
dious, with just a few interesting cases. Their complexity arises from the man-
agement of many details rather than from deep conceptual di�culties; yet small
mistakes or overlooked cases can invalidate large amounts of work. These e�ects
are ampli�ed as languages scale: it becomes hard to keep de�nitions and proofs
consistent, to reuse work, and to ensure tight relationships between theory and
implementations. Automated proof assistants o�er the hope of signi�cantly eas-
ing these problems. However, despite much encouraging progress in recent years
and the availability of several mature tools, their use is still not commonplace.

The challenge itself consists of formalizing the operational semantics and the type system
of F<:, a typed functional language featuring polymorphism and subtyping, and proving the
soundness of the type system with respect to this semantics. As a guidance, the statement of
the challenge provides a detailed on-paper formalization of F<: and a proof of type soundness
written in ordinary mathematics. The challenge itself is to express these formalization and
proofs in a theorem prover in such a way that the proofs can be mechanically checked.

A dozen complete or partial solutions to the POPLmark challenge were developed, using
a wide variety of proof assistants (Coq, Isabelle/HOL, Twelf, . . . ) and of representation
techniques for the terms, types and rules of the F<: language. These solutions can be found

4 Xavier Leroy

on the POPLmark Web site1. This report presents the solution that we developed. It uses
the Coq proof assistant [Coq07, BC04], as well as the so-called �locally nameless� represen-
tation introduced by McKinna, Pollack, Gordon, and McBride, [MP99, Gor94, MM04]. The
complete Coq development is available online [Ler07].

1.2 Locally nameless representations

Perhaps surprisingly, the main di�culty in the POPLmark challenge is not to translate the
high-level, on-paper proofs into equivalent mechanized proofs. These on-paper proofs are
mostly syntactic in nature and do not involve higher mathematics. Such proofs are handled
well by many existing proof assistants.

What makes POPLmark di�cult is the need to correctly handle binders, bound variables
and alpha-conversion. The terms and types of F<: contain variables such as function pa-
rameters and type variables, as well as constructs that bind these variables, such as function
abstractions λx. x in terms and universal quanti�cation ∀X. X → X in types. To obtain a
sound theory of the language, it is necessary to treat terms and types as equal up to alpha-
conversion, that is, renamings of bound variables. For instance, the two types ∀X. X → X
and ∀Y. Y → Y must be treated as equal. This is di�cult to achieve with today's proof
assistants, because these o�er poor support for quotient sets. Several representations of
binders have been investigated over the last 40 years to overcome this di�culty.

Nominal representations follow the usual mathematical practice of identifying bound
variables by names and quotienting terms up to alpha-conversion of bound names. This
can be internalized within the logic itself, leading to nominal logics [Pit03]. Support for
a nominal logic within a proof assistant is still in its infancy, but is making signi�cant
progress [UT05]. Despite these di�culties, this approach is attractive because it is the
closest to usual mathematical practice, leading to statements and proofs that are very
close to what we are accustomed to see in textbooks and research papers.

de Bruijn indices avoid the issue with alpha-conversion by representing variables not by
names, but by position (indices) relative to the enclosing binders: v1 is the variable
bound by the �rst enclosing binder, v2 by the second, etc. For instance, the term
λx. x (λy. x y) in nominal notation is represented as λ. v1 (λ. v2 v1) in de Bruijn
notation. The strength of this approach is that terms have unique representations:
two terms are alpha-equivalent in nominal notation if and only if they are identical in
de Bruijn notation. It is therefore easy to represent and work with in a proof assistant
[Hue94]. The downside of this approach is that the meaning of de Bruijn indices is
very dependent on the context. Many lifting and relocation operations over indices
must be included in the statements of theorems, making them unnatural and hard to
read.

1http://fling-l.seas.upenn.edu/~plclub/cgi-bin/poplmark/

Higher-order abstract syntax (HOAS) uses the functions provided by the logic to rep-
resent binders. For example, we have the following representations for the term λx.x:

Style Representation of λx.x
nominal Lambda("x", Var "x")

de Bruijn Lambda(Var 1)

HOAS Lambda(fun x -> x)

(The fun construct denotes a function of the logic, binding a logical variable x that
ranges over all terms.) The beauty of this approach is that alpha-conversion and
substitution of bound names are handled automatically by the logic and need not be
managed explicitly. The downside of HOAS is that it is not compatible with the rich
logics of general-purpose proof assistants such as Coq and Isabelle/HOL and therefore
can only be supported by systems such as Twelf's metatheory, which are less expressive.
Also, it leads to statements of theorems that are quite di�erent from what we write in
ordinary mathematics.

In our solution to the POPLmark challenge, we use a representation for binders that
is a combination of the nominal approach with de Bruijn indices. This representation is
known as the �locally nameless� approach [MP99, Gor94, MM04]. It uses de Bruijn indices
to identify bound variables, and names to identify free variables. For example, we have the
following representations:

Style Representation of λx. y x
nominal Lambda("x", App(Var "y", Var "x"))

de Bruijn Lambda(App(Var 2, Var 1))

locally nameless Lambda(App(Freevar "y", Boundvar 1))

Like de Bruijn indices, the locally nameless representation has the very nice property that
alpha-equivalent terms have syntactically equal representations. Like nominal approaches,
the locally nameless representation enables us to use familiar names to refer to free variables,
leading to statements of theorems that are close to ordinary mathematics.

A crucial invariant in the locally nameless approach is that representations of terms never
contain free de Bruijn indices. Consequently, when recursing over a de Bruijn-closed term
and encountering a binder such as Lambda(t), it is incorrect to recurse over t, since this
term can contain Boundvar 1 as a free de Bruijn index. The correct approach is to invent
a fresh name x to stand for the formal parameter of the lambda-abstraction and recurse
over t[0 ← x], that is, the term obtained by substituting the free variable Freevar "x"

for all occurrences of the bound variable Boundvar 1 in t. (See section 2.2.3 for detailed
explanations.) This style of de�nition is slightly unnatural at �rst, but is reasonably easy
to get used to. In particular, this substitution by a fresh name materializes the so-called
�Barendregt convention� for nominal terms.

RR n° 6098



6 Xavier Leroy

1.3 Outline

The remainder of this report is a complete step-by-step presentation of our Coq development
in literate programming style. All de�nitions, theorems and intermediate lemmas are shown
in Coq syntax, interspersed with explanations. We omit the proof scripts for all lemmas
but show them for the main theorems, so that the reader can get a feeling of the size and
complexity of the proof scripts. The full Coq development, including scripts, is available
online [Ler07].

� Chapter 2 de�nes the syntax of F<: type expressions and the subtyping relation be-
tween types, and proves three key properties of this relation: re�exivity, transitivity,
and stability by substitution. It corresponds to part 1A and a fragment of part 2A of
the POPLmark challenge.

� Chapter 3 de�nes the syntax of terms, the type system, and the dynamic semantics of
F<:. It proves the soundness of the type system with respect to the semantics. This
corresponds to the remainder of part 2A of the challenge.

� Chapter 4 shows how executions of F<: programs can be performed (for testing pur-
poses), both within Coq and through �extraction� (automatic generation) of Caml
code for a reference F<: interpreter. This is part 3 of the challenge.

� Chapter 5 concludes this report by an informal assessment of the quality of our solution,
and more generally of the usability of the �locally nameless� approach and of the Coq
proof assistant for POPLmark-style problems.

INRIA



A locally nameless solution to the POPLmark challenge 7

Chapter 2

Algorithmic subtyping

This chapter corresponds to part 1A of the POPLmark challenge, namely the formal def-
inition of a subtyping relation between types of F<: and the proof of basic type-theoretic
properties of this relation.

We start by �importing� three modules from the Coq library of standard de�nitions and
theorems: Arith (arithmetic over natural numbers), ZArith (arithmetic over integers) and
List (operations over �nite lists). We also import a module extralibrary that we developed
specially, which provides additional lemmas about list membership. This module is omitted
in this report, but available as part of the Web version of this development [Ler07].

Require Import Arith.
Require Import ZArith.
Require Import List.
Require Import extralibrary.

2.1 Names and swaps of names

We use names (also called atoms) to represent free variables in terms. Any in�nite type with
decidable equality will do. In preparation for the second part of the challenge, we attach a
kind to every name: either �type� or �term�, and ensure that there are in�nitely many names
of each kind. Concretely, we represent names by pairs of a kind and an integer (type Z).

Inductive name kind : Set :=
| TYPE : name kind
| TERM : name kind.

De�nition name : Set := (name kind × Z )%type.

De�nition kind (n: name) : name kind := fst n.

RR n° 6098



8 Xavier Leroy

Equality between names is decidable.

Lemma eq name: ∀ (n1 n2 : name), {n1 = n2} + {n1 6= n2}.

Moreover, we have the following obvious simpli�cation rules on tests over name equality.

Lemma eq name true:
∀ (A: Set) (n: name) (a b: A), (if eq name n n then a else b) = a.

Lemma eq name false:
∀ (A: Set) (n m: name) (a b: A), n 6= m → (if eq name n m then a else b) = b.

The following lemma shows that there always exists a name of the given kind that is fresh
w.r.t. the given list of names, that is, distinct from all the names in this list.

De�nition �nd fresh name (k : name kind) (l : list name) : name :=
(k, 1 + fold right (fun (n:name) x ⇒ Zmax (snd n) x ) 0 l)%Z.

Lemma �nd fresh name is fresh:
∀ k l, let n := �nd fresh name k l in ¬In n l ∧ kind n = k.

Lemma fresh name:
∀ (k : name kind) (l : list name), ∃ n, ¬In n l ∧ kind n = k.

As argued by Pitts and others, swaps (permutations of two names) are an interesting special
case of renamings. We will use swaps later to prove that our de�nitions are equivariant,
that is, insensitive to the choices of fresh identi�ers.

De�nition swap (u v x : name) : name :=
if eq name x u then v else if eq name x v then u else x.

The following lemmas are standard properties of swaps: self-inverse, injective, kind-preserving.

Lemma swap left : ∀ x y, swap x y x = y.

Lemma swap other : ∀ x y z, z 6= x → z 6= y → swap x y z = z.

Lemma swap inj : ∀ u v x y, swap u v x = swap u v y → x = y.

Lemma swap kind : ∀ u v x, kind u = kind v → kind (swap u v x ) = kind x.

2.2 Types and typing environments

2.2.1 Type expressions

The syntax of type expressions is standard, except that we have two representations for
variables: Tparam represents free type variables, identi�ed by a name, while Tvar represents

INRIA

bound type variables, identi�ed by their de Bruijn indices. Our de Bruijn indices start at 0.
In a Forall t1 t2 type, the variable Tvar 0 is bound by the Forall in type t2.

Inductive type: Set :=
| Tparam: name → type
| Tvar : nat → type
| Top: type
| Arrow : type → type → type
| Forall : type → type → type.

The free names of a type are de�ned as follow. Notice the Forall case: Forall does not bind
any name.

Fixpoint fv type (t : type) : list name :=
match t with
| Tparam x ⇒ x :: nil
| Tvar n ⇒ nil
| Top ⇒ nil
| Arrow t1 t2 ⇒ fv type t1 ++ fv type t2
| Forall t1 t2 ⇒ fv type t1 ++ fv type t2
end.

There are two substitution operations over types, written vsubst and psubst in Pollack's
talk. vsubst substitutes a type for a bound variable (a de Bruijn index). psubst substitutes
a type for a free variable (a name).

The crucial observation is that variable capture cannot occur during either substitution:

� Types never contain free de Bruijn indices, since these indices are used only for rep-
resenting bound variables. Therefore, vsubst does not need to perform lifting of de
Bruijn indices in the substituted type.

� Types never bind names, only de Bruijn indices. Therefore, psubst never needs to
perform renaming of names in the substituted term when descending below a binder.

Fixpoint vsubst type (a: type) (x : nat) (b: type) {struct a} : type :=
match a with
| Tparam n ⇒ Tparam n
| Tvar n ⇒

match compare nat n x with
| Nat less ⇒ Tvar n
| Nat equal ⇒ b
| Nat greater ⇒ Tvar (pred n)
end

| Top ⇒ Top
| Arrow a1 a2 ⇒ Arrow (vsubst type a1 x b) (vsubst type a2 x b)

| Forall a1 a2 ⇒ Forall (vsubst type a1 x b) (vsubst type a2 (S x ) b)
end.

Fixpoint psubst type (a: type) (x : name) (b: type) {struct a} : type :=
match a with
| Tparam n ⇒ if eq name n x then b else Tparam n
| Tvar n ⇒ Tvar n
| Top ⇒ Top
| Arrow a1 a2 ⇒ Arrow (psubst type a1 x b) (psubst type a2 x b)
| Forall a1 a2 ⇒ Forall (psubst type a1 x b) (psubst type a2 x b)
end.

In the remainder of the development, vsubst is only used to replace bound variable 0 by a
fresh, free variable (a name) when taking apart a Forall type. This operation is similar to
the �freshening� operation used in Fresh ML and related systems. We call it freshen type
for clarity.

De�nition freshen type (a: type) (x : name) : type := vsubst type a 0 (Tparam x ).

Free variables and freshening play well together.

Lemma fv type vsubst type:
∀ x a n b, In x (fv type a) → In x (fv type (vsubst type a n b)).

Lemma fv type freshen type:
∀ x a y, In x (fv type a) → In x (fv type (freshen type a y)).

We now de�ne swaps (permutation of names) over types and show basic properties of swaps
that will be useful later.

Fixpoint swap type (u v : name) (t : type) {struct t} : type :=
match t with
| Tparam x ⇒ Tparam (swap u v x )
| Tvar n ⇒ Tvar n
| Top ⇒ Top
| Arrow t1 t2 ⇒ Arrow (swap type u v t1 ) (swap type u v t2 )
| Forall t1 t2 ⇒ Forall (swap type u v t1 ) (swap type u v t2 )
end.

Swaps are involutions (self-inverse).

Lemma swap type inv : ∀ u v t, swap type u v (swap type u v t) = t.

Swaps of variables that do not occur free in a type leave the type unchanged.

Lemma swap type not free:
∀ u v t, ¬In u (fv type t) → ¬In v (fv type t) → swap type u v t = t.

Swaps commute with vsubst substitution and freshening.

INRIA

Lemma vsubst type swap:
∀ u v a n b,
swap type u v (vsubst type a n b) = vsubst type (swap type u v a) n (swap type u v b).

Lemma freshen type swap:
∀ u v a x,
swap type u v (freshen type a x ) = freshen type (swap type u v a) (swap u v x ).

Swaps commute with the computation of free variables.

Lemma in fv type swap:
∀ u v x t, In x (fv type t) ↔ In (swap u v x ) (fv type (swap type u v t)).

2.2.2 Typing environments

Typing environments are standard: lists of (name, type) pairs. Bindings are added to the
left of the environment using the cons list operation. Thus, later bindings come �rst.

De�nition typenv := list (name × type).

De�nition dom (e: typenv) := map (@fst name type) e.

Lemma dom append : ∀ e1 e2, dom (e1 ++ e2 ) = dom e1 ++ dom e2.

The lookup function returns the type associated with a name in a typing environment.

Fixpoint lookup (x : name) (e: typenv) {struct e} : option type :=
match e with
| nil ⇒ None
| (y, t) :: e' ⇒ if eq name x y then Some t else lookup x e'
end.

Lemma lookup inv : ∀ x t e, lookup x e = Some t → In x (dom e).

Lemma lookup exists: ∀ x e, In x (dom e) → ∃ t, lookup x e = Some t.

We extend swaps to typing environments, pointwise.

Fixpoint swap env (u v : name) (e: typenv) {struct e} : typenv :=
match e with
| nil ⇒ nil
| (x, t) :: e' ⇒ (swap u v x, swap type u v t) :: swap env u v e'
end.

Environment lookup commutes with swaps.

Lemma lookup swap:
∀ u v x e t, lookup x e = Some t →
lookup (swap u v x ) (swap env u v e) = Some (swap type u v t).

The dom operation commutes with swaps.

Lemma in dom swap:
∀ u v x e, In x (dom e) ↔ In (swap u v x ) (dom (swap env u v e)).

2.2.3 Well-formedness of types and environments

A type is well-formed in a typing environment if:

� all names free in the type are of kind TYPE ;

� all names free in the type are bound in the environment;

� it does not contain free de Bruijn variables.

We capture these conditions by the following inference rules.

Inductive wf type: typenv → type → Prop :=
| wf type param: ∀ x e,

kind x = TYPE → In x (dom e) →
wf type e (Tparam x )

| wf type top: ∀ e,
wf type e Top

| wf type arrow : ∀ e t1 t2,
wf type e t1 → wf type e t2 → wf type e (Arrow t1 t2 )

| wf type forall : ∀ e t1 t2,
wf type e t1 →
(∀ x,

kind x = TYPE → ¬In x (fv type t2 ) → ¬In x (dom e) →
wf type ((x, t1 ) :: e) (freshen type t2 x )) →

wf type e (Forall t1 t2 ).

The rules are straightforward, except perhaps the wf type forall rule. It follows a general
pattern for operating over sub-terms of a binder, such as t2 in Forall t1 t2. The de Bruijn
variable Tvar 0 is potentially free in t2. To recover a well-formed term, without free de
Bruijn variables, we substitute Tvar 0 with a fresh name x. Therefore, the premise for t2
applies to freshen type t2 x.

How should x be chosen? As in the name-based speci�cation, x must not be in the domain
of e, otherwise the extended environment (x, t1 ) :: e would be ill-formed. In addition, the
name x must not be free in t2, otherwise the freshening freshen type t2 x would incorrectly
identify the bound, universally-quanti�ed variable of the Forall types with an existing, free
type variable.

How should x be quanti�ed? That is, should the second premise wf type ((x, t1 ) :: e)
(freshen type t2 x )) hold for one particular name x not in dom(e), or for all names x not in

INRIA

dom(e)? The �for all� alternative obviously leads to a stronger induction principle: proofs
that proceed by inversion or induction over an hypothesis wf type e (Forall t1 t2 ) can then
choose any convenient x fresh for e to exploit the second premise, rather than having to cope
with a �xed, earlier choice of x. Symmetrically, the �for one� alternative is more convenient
for proofs that must conclude wf type e (Forall t1 t2 ): it su�ces to exhibit one suitable
x fresh in e that satis�es the second premise, rather than having to establish the second
premise for all such x.

The crucial observation is that those two alternative are equivalent: the same subtyping
judgements can be derived with the �for all� rule and the �for one� rule. (See Pollack's talk
for more explanations.) Therefore, in the de�nition of the wf type predicate above, we chose
the �for all� rule, so as to get the strongest induction principle. And we will show shortly
that the �for one� rule is admissible and can be used in proofs that conclude wf type e (Forall
t1 t2 ).

An environment is well-formed if every type it contains is well-formed in the part of the
environment that occurs to its right, i.e. the environment at the time this type was intro-
duced. This ensures in particular that all the variables in this type are bound earlier (i.e.
to the right) in the environment. Moreover, we impose that no name is bound twice in an
environment.

Inductive wf env : typenv → Prop :=
| wf env nil :

wf env nil
| wf env cons: ∀ x t e,

wf env e → ¬In x (dom e) → wf type e t →
wf env ((x, t) :: e).

Lemma wf type env incr :
∀ e t, wf type e t → ∀ e', incl (dom e) (dom e' ) → wf type e' t.

A type well formed in e has all its free names in the domain of e.

Lemma fv wf type: ∀ x e t, wf type e t → In x (fv type t) → In x (dom e).

Looking up the type of a name in a well-formed environment returns a well-formed type.

Lemma wf type lookup: ∀ e, wf env e → ∀ x t, lookup x e = Some t → wf type e t.

Type well-formedness is stable by swapping.

Lemma wf type swap:
∀ u v e t,
kind u = kind v → wf type e t → wf type (swap env u v e) (swap type u v t).

Environment well-formedness is stable by swapping.

Lemma wf env swap:
∀ u v e, kind u = kind v → wf env e → wf env (swap env u v e).

The domain of an environment is invariant by swaps of names that are not in this domain.

Lemma swap env dom:
∀ u v e, ¬In u (dom e) → ¬In v (dom e) → dom (swap env u v e) = dom e.

A well-formed environment is invariant by swaps of names that are not in the domains of
this environment.

Lemma swap env not free:
∀ u v e, wf env e → ¬In u (dom e) → ¬In v (dom e) → swap env u v e = e.

We now show that the alternate formulation of rule wf type forall (the one with �for one
fresh name x � instead of �for all fresh names x � in the second premise) is admissible.

Lemma wf type forall' :
∀ e x t t1 t2,
wf type e t1 → kind x = TYPE → ¬In x (fv type t2 ) → ¬In x (dom e) →
wf type ((x, t) :: e) (freshen type t2 x ) →
wf type e (Forall t1 t2 ).

2.3 Algorithmic subtyping

We now de�ne the subtyping judgement as an inductive predicate. Each constructor of the
predicate corresponds to an inference rule in the original de�nition of subtyping.

Inductive is subtype: typenv → type → type → Prop :=
| sa top: ∀ e s,

wf env e → wf type e s →
is subtype e s Top

| sa re� tvar : ∀ e x u,
wf env e → kind x = TYPE → lookup x e = Some u →
is subtype e (Tparam x ) (Tparam x )

| sa trans tvar : ∀ e x u t,
kind x = TYPE → lookup x e = Some u → is subtype e u t →
is subtype e (Tparam x ) t

| sa arrow : ∀ e s1 s2 t1 t2,
is subtype e t1 s1 → is subtype e s2 t2 →
is subtype e (Arrow s1 s2 ) (Arrow t1 t2 )

| sa all : ∀ e s1 s2 t1 t2,
is subtype e t1 s1 →
(∀ x,

kind x = TYPE → ¬In x (dom e) →
is subtype ((x, t1 ) :: e) (freshen type s2 x ) (freshen type t2 x )) →

is subtype e (Forall s1 s2 ) (Forall t1 t2 ).

INRIA

The sa all rule for Forall types follows the pattern that we already introduced for the
wf type predicate, in rule wf type forall. In the original, name-based speci�cation, we say
that E ` (∀x <: σ1. σ2) <: (∀x <: τ1. τ2) if E ` τ1 <: σ1 and E, x : τ1 ` σ2 <: τ2. The type
variable x, being α-convertible in the conclusion, is (implicitly or explicitly) chosen so that
E, x : τ1 is well-formed in the second premise, that is, x is chosen not free in E.

In our locally nameless representation, the type variables bound by Forall in the con-
clusion do not have names. We must therefore invent a suitable name x and substitute it
for the bound variable TVar 0 in the types s2 and t2. Therefore, the second premise puts
freshen type s2 x and freshen type t2 x in subtype relation.

As mentioned already, x should be chosen not in the domain of e (otherwise the ex-
tended environment (x, t1 ) :: e would be ill-formed) and not free in s2 and t2, otherwise the
freshenings freshen type s2 x and freshen type t2 x would incorrectly identify the bound,
universally-quanti�ed variable of the Forall types with an existing, free type variable. How-
ever, as we will prove below, the rules for is subtype satisfy a well-formedness condition: if
is subtype e u1 u2, then u1 and u2 are well-formed in e, implying that a name not in the
domain of e cannot be free in u1 or u2. Therefore, the condition �x not in the domain of e�
su�ces to ensure that x is not free in s2 and t2, and that the freshenings freshen type s2 x
and freshen type t2 x make sense.

As mentioned already as well, we have a choice between quantifying over all suitable x
or over one suitable x in the second premise. Again, we go with the �for all� alternative in
order to obtain the strongest induction principle, and we will show later that the �for one�
alternative is derivable.

For the time being, we start with simple well-formedness properties of the types and
environments involved in a is subtype relation.

Lemma is subtype wf env : ∀ e s t, is subtype e s t → wf env e.

Lemma is subtype wf type: ∀ e s t, is subtype e s t → wf type e s ∧ wf type e t.

Lemma is subtype wf type l : ∀ e s t, is subtype e s t → wf type e s.

Lemma is subtype wf type r : ∀ e s t, is subtype e s t → wf type e t.

We now show that the is subtype predicate is stable by swapping. This property is crucial
to show the equivalence of the �for all� and �for one� interpretations of rule sa all.

Lemma is subtype swap:
∀ u v, kind u = kind v →
∀ e s t, is subtype e s t →
is subtype (swap env u v e) (swap type u v s) (swap type u v t).

Two silly lemmas about freshness of names in types.

Lemma fresh wf type: ∀ x e t, wf type e t → ¬In x (dom e) → ¬In x (fv type t).

Lemma fresh freshen type:
∀ x t1 e t y,
wf type ((x, t1 ) :: e) (freshen type t x ) → ¬In y (dom e) → x 6= y → ¬In y (fv type t).

We now show that the alternate presentation of rule sa all (the one with �for one name� in
the second premise instead of �for all names�) is admissible.

Lemma sa all' :
∀ e s1 s2 t1 t2 x,
is subtype e t1 s1 →
kind x = TYPE → ¬In x (dom e) → ¬In x (fv type s2 ) → ¬In x (fv type t2 ) →
is subtype ((x, t1 ) :: e) (freshen type s2 x ) (freshen type t2 x ) →
is subtype e (Forall s1 s2 ) (Forall t1 t2 ).

2.4 Re�exivity and transitivity of subtyping

We now turn (at last!) to proving the two theorems of part 1 of the POPLmark challenge:
re�exivity and transitivity of subtyping.

Re�exivity of subtyping is shown by straightforward induction on the derivation of well-
formedness of the type. As noted by McKinna and Pollack, such inductions conveniently
replace inductions on the structure of types.

Theorem sub re� : ∀ t e, wf type e t → wf env e → is subtype e t t.
Proof.
induction 1; intros.

Case t = Tparam x
destruct (lookup exists H0 ) as [t L].
apply sa re� tvar with t ; auto.

Case t = Top
apply sa top. auto. constructor.

Case t = Arrow t1 t2
apply sa arrow. auto. auto.

Case t = Forall t1 t2
destruct (fresh name TYPE (fv type t2 ++ dom e)) as [x [FRESH KIND]].
apply sa all' with x ; eauto.
apply H1 ; eauto.
constructor ; eauto.

Qed.

We now do some sca�olding work for the proof of transitivity. First, we will need to perform
inductions over the size of types. We cannot just do inductions over the structure of types,
as the original paper proof did, because in the case of Forall t1 t2, we will need to recurse

not on t2 but on freshen type t2 x for some x, which is not a sub-term of t2. However, the
size of freshen type t2 x is the same as the size of t2, so induction over sizes will work.

Fixpoint size type (t : type): nat :=
match t with
| Tparam ⇒ 0
| Tvar ⇒ 0
| Top ⇒ 0
| Arrow t1 t2 ⇒ 1 + size type t1 + size type t2
| Forall t1 t2 ⇒ 1 + size type t1 + size type t2
end.

Lemma vsubst type size: ∀ x a n, size type (vsubst type a n (Tparam x )) = size type a.

Lemma freshen type size: ∀ x a, size type (freshen type a x ) = size type a.

We now de�ne a notion of inclusion between environments that we call �weakening�. e2
weakens e1 if all the bindings in e1 are preserved in e2 ; however, e2 may contain additional,
non-interfering bindings.

De�nition env weaken (e1 e2 : typenv) : Prop :=
∀ x t, lookup x e1 = Some t → lookup x e2 = Some t.

Lemma env weaken incl dom: ∀ e1 e2, env weaken e1 e2 → incl (dom e1 ) (dom e2 ).

The subtyping relation is stable by weakening of the typing environment.

Lemma sub weaken:
∀ e s t, is subtype e s t → ∀ e', wf env e' → env weaken e e' → is subtype e' s t.

A special case of weakening is the addition of bindings to an existing typing environment,
provided the resulting environment is well-formed.

Lemma env concat weaken:
∀ delta gamma,
wf env (delta ++ gamma) → env weaken gamma (delta ++ gamma).

The following lemmas prove useful properties of environments of the form e1 ++ (x, p) ::
e2, that is, all bindings of e2, followed by a binding of p to x, followed by all bindings of e1.

Lemma dom env extends: ∀ e1 x p q e2, dom (e1 ++ (x, p) :: e2 ) = dom (e1 ++ (x, q)
:: e2 ).

Lemma wf env extends:
∀ e2 x p q e1, wf env (e1 ++ (x, p) :: e2 ) → wf type e2 q → wf env (e1 ++ (x, q) ::

e2 ).

Lemma lookup env extends:
∀ e2 x p q y e1,
wf env (e1 ++ (x, q) :: e2 ) →

lookup y (e1 ++ (x, p) :: e2 ) = if eq name y x then Some p else lookup y (e1 ++ (x,
q) :: e2 ).

Now comes the major result: transitivity and the narrowing property of subtyping, proved
simultaneously. The proof follows the structure of the paper proof, with the structural
induction on q being replaced by a Peano induction on the size of q.

Lemma sub trans narrow :
∀ n,

(∀ e s q t,
size type q ≤ n →
is subtype e s q → is subtype e q t →
is subtype e s t)

∧ (∀ x e1 e2 p q r s,
size type q ≤ n →
is subtype (e1 ++ (x, q) :: e2 ) r s → is subtype e2 p q →
is subtype (e1 ++ (x, p) :: e2 ) r s).

Proof.
intro n0. pattern n0. apply Peano induction.
intros size HRsize.

Part 1: transitivity
assert (∀ e s q, is subtype e s q →

∀ t, size type q ≤ size → is subtype e q t → is subtype e s t).
Sub-induction on the derivation of is subtype e s q
induction 1; intros.

Case sa top
inversion H2. apply sa top. auto. eauto.

Case sa re� tvar
auto.

Case sa trans tvar
apply sa trans tvar with u; auto.

Case sa arrow
inversion H2.
apply sa top. auto. inversion H4. constructor ; eauto.
subst e0 ; subst s0 ; subst s3. simpl in H1.
assert (SZpos: pred size < size). omega.
elim (HRsize (pred size) SZpos); intros HR1 HR2.
apply sa arrow.

Application of the outer induction hypothesis to t1
apply HR1 with t1 ; auto. omega.

Application of the outer induction hypothesis to t2
apply HR1 with t2 ; auto. omega.

Case sa forall
inversion H3.

apply sa top. auto.
apply is subtype wf type l with (Forall t1 t2 ). constructor ; assumption.
subst e0 ; subst s0 ; subst s3. simpl in H2.
assert (SZpos: pred size < size). omega.
elim (HRsize (pred size) SZpos); intros HR1 HR2.

Choice of an appropriately fresh name x
elim (fresh name TYPE (dom e ++ fv type t3 ++ fv type s2 )).
intros x [FRESH KIND].
apply sa all' with x ; eauto.

Application of the outer induction hypothesis to t1
apply HR1 with t1 ; auto. omega.

Application of the outer induction hypothesis to freshen t2 x
apply HR1 with (freshen type t2 x ).
rewrite freshen type size. omega.
change ((x, t0 ) :: e) with (nil ++ (x, t0 ) :: e).

Application of the narrowing part of the outer induction hypothesis.
apply HR2 with t1. omega.
simpl. apply H0 ; eauto. auto. apply H9 ; eauto.

Part 2: narrowing
assert (∀ e r s,
is subtype e r s →
∀ e1 x q e2 p,
e = e1 ++ (x, q) :: e2 → size type q ≤ size → is subtype e2 p q →
is subtype (e1 ++ (x, p) :: e2 ) r s).

Sub-induction on the derivation of is subtype e r s
induction 1; intros; subst e.

Case sa top
apply sa top. apply wf env extends with q ; eauto.
apply wf type env incr with (e1 ++ (x, q) :: e2 ); auto.
rewrite (dom env extends e1 x q p e2 ). apply incl re�.

Case sa re� tvar
apply sa re� tvar with (if eq name x x0 then p else u).
apply wf env extends with q ; eauto. auto.
rewrite (lookup env extends e2 x0 p q x e1 H0 ).
case (eq name x x0 ); auto.

Case sa trans tvar
apply sa trans tvar with (if eq name x x0 then p else u).
auto. rewrite (lookup env extends e2 x0 p q x e1 ).
case (eq name x x0 ); auto. eauto.
case (eq name x x0 ); intro.

sub-case x = x0
generalize H1. rewrite (lookup env extends e2 x0 q q x e1 ); eauto.

rewrite e; rewrite eq name true; intro EQ ; injection EQ ; intro; subst u.
apply H with q.
apply sub weaken with e2 ; auto.
apply wf env extends with q ; eauto.
change (e1 ++ (x0, p) :: e2 ) with (e1 ++ (((x0, p) :: nil) ++ e2 )).
rewrite ← app ass. apply env concat weaken.
rewrite app ass. simpl. apply wf env extends with q ; eauto.
auto.
apply IHis subtype with q ; auto.

sub-case x 6= x0
apply IHis subtype with q ; auto.

Case sa arrow
apply sa arrow.
apply IHis subtype1 with q ; auto.
apply IHis subtype2 with q ; auto.

Case sa forall
apply sa all.
apply IHis subtype with q ; auto.
intros.
change ((x0, t1 ) :: e1 ++ (x, p) :: e2 )
with (((x0, t1 ) :: e1 ) ++ (x, p) :: e2 ).

apply H2 with q. auto.
rewrite (dom env extends e1 x q p e2 ). auto.
re�exivity. auto. auto.

Combining the two parts together
split. intros; apply H with q ; auto.
intros; apply H0 with (e1 ++ (x, q) :: e2 ) q ; auto.

Qed.

As a corollary, we obtain transitivity of subtyping.

Theorem sub trans:
∀ e s q t, is subtype e s q → is subtype e q t → is subtype e s t.

As well as narrowing.

Theorem sub narrow :
∀ x e1 e2 p q r s,
is subtype (e1 ++ (x, q) :: e2 ) r s → is subtype e2 p q →
is subtype (e1 ++ (x, p) :: e2 ) r s.

2.5 Stability of the subtyping judgement under substitu-

tions

We now prove a property of the subtyping judgement that plays a crucial role later to prove
that instantiation of polymorphic types is semantically sound. This property is stability
under substitutions, namely that if Γ, X <: U ` T1 <: T2, then it must be the case that
Γ ` T1[X ← U ] <: T2[X ← U ]. In the POPLmark challenge, this property belongs to
part 2 of the challenge, but we prefer to prove it now, since it involves only the subtyping
judgement.

To prove stability under substitution by induction, we need to strengthen its statement
as follows: if Γ, X <: U,Γ′ ` T1 <: T2, then Γ,Γ′[X ← U ] ` T1[X ← U ] <: T2[X ← U ]. We
therefore need to extend type substitutions to environments, pointwise.

Fixpoint psubst env (e: typenv) (x : name) (b: type) {struct e} : typenv :=
match e with
| nil ⇒ nil
| (y, t) :: e' ⇒ (y, psubst type t x b) :: psubst env e' x b
end.

Lemma dom psubst env : ∀ x b e, dom (psubst env e x b) = dom e.

2.5.1 Commutation properties for type substitutions

As a preliminary result, we need to show that well-formedness of types and environments
is preserved by substitution of a well-formed type for a name. This simple property needs
a number of commutation properties between the psubst type, freshen type and vsubst type
functions. Unfortunately, some hacking on de Bruijn indices is necessary. We start by
de�ning a predicate type vars below t n that holds if all free de Bruijn variables in the
type t are less than n. In particular, type vars below t 0 holds i� t has no free de Bruijn
variables.

Fixpoint type vars below (t : type) (n: nat) {struct t} : Prop :=
match t with
| Tparam x ⇒ True
| Tvar m ⇒ m < n
| Top ⇒ True
| Arrow t1 t2 ⇒ type vars below t1 n ∧ type vars below t2 n
| Forall t1 t2 ⇒ type vars below t1 n ∧ type vars below t2 (S n)
end.

Lemma type vars below vsubst :
∀ t n t', type vars below (vsubst type t n t' ) n → type vars below t (S n).

A well-formed type, having no free de Bruijn variables, has all its de Bruijn variables below
0.

Lemma wf type vars below 0 : ∀ e t, wf type e t → type vars below t 0.

A type is invariant by vsubst type substitution if the de Bruijn variable being substituted is
not free in the type.

Lemma vsubst invariant below :
∀ t n m s, type vars below t n → n ≤ m → vsubst type t m s = t.

As a corollary, well-formed types are invariant by vsubst substitutions.

Lemma vsubst wf type: ∀ e t n s, wf type e t → vsubst type t n s = t.

It follows that vsubst and psubst substitutions commute in the following sense.

Lemma psubst vsubst type:
∀ e a x b n c,
wf type e b →
vsubst type (psubst type a x b) n (psubst type c x b) =
psubst type (vsubst type a n c) x b.

Consequently, psubst type and freshen type commute if they operate on distinct names.

Lemma psubst freshen type:
∀ e a x b y,
wf type e b → x 6= y →
freshen type (psubst type a x b) y = psubst type (freshen type a y) x b.

Additionally, vsubst substitution of de Bruijn variable 0 is equivalent to freshening with a
fresh type name x followed by a psubst substitution over x.

Lemma vsubst psubst type:
∀ x t2 t1 n, ¬In x (fv type t1 ) →
vsubst type t1 n t2 = psubst type (vsubst type t1 n (Tparam x )) x t2.

Lemma vsubst psubst freshen type:
∀ x t1 t2, ¬In x (fv type t1 ) →
vsubst type t1 0 t2 = psubst type (freshen type t1 x ) x t2.

2.5.2 Preservation of well-formedness of types and environments

during substitution

Well-formedness of types is preserved by psubst type substitution of a well-formed type for
a type name.

Lemma wf type psubst :

∀ e1 x p q e2 t,
wf type (e2 ++ (x, p) :: e1 ) t → wf type e1 q →
wf type (psubst env e2 x q ++ e1 ) (psubst type t x q).

Similarly, well-formedness of environments is preserved by psubst env substitution.

Lemma wf env psubst :
∀ e1 x p q e2,
wf env (e2 ++ (x, p) :: e1 ) → wf type e1 q →
wf env (psubst env e2 x q ++ e1 ).

2.5.3 Type substitution preserves subtyping

We now show that if s is subtype of t in the environment e2 ++ (x, p) :: e1, and if q is
subtype of p in e1, then psubst s x q is subtype of psubst t x q in psubst env x q e2 ++ e1.

Well-formed environments of the form e2 ++ (x, p) :: e1 are such that x is not in the
domain of e1 and not free in the types listed in e1. Therefore, these types are invariant by
substitution over x.

Lemma env concat not free: ∀ e1 x p e2, wf env (e2 ++ (x, p) :: e1 ) → ¬In x (dom e1 ).

Lemma psubst type inv : ∀ t x s, ¬In x (fv type t) → psubst type t x s = t.

The following technical lemma relates lookup operations between the original environment
e2 ++ (x, p) :: e1 and its substituted counterpart, psubst env e2 x q ++ e1.

Lemma lookup env concat :
∀ e1 x p q y t e2,
wf env (e2 ++ (x, p) :: e1 ) →
lookup y (e2 ++ (x, p) :: e1 ) = Some t →
(x = y ∧ t = p) ∨
(x 6= y ∧ lookup y (psubst env e2 x q ++ e1 ) = Some (psubst type t x q)).

We can now prove the stability of subtyping by type substitution (lemma A.10 in the chal-
lenge statement). The proof proceeds by induction over the derivation of the second sub-
typing hypothesis.

Theorem sub stable subst :
∀ e1 x p q e2 s t,
is subtype e1 p q →
is subtype (e2 ++ (x, q) :: e1 ) s t →
is subtype (psubst env e2 x p ++ e1 ) (psubst type s x p) (psubst type t x p).

Proof.
assert (∀ e1 x p q, is subtype e1 p q →

∀ e s t, is subtype e s t →

∀ e2, e = e2 ++ (x, q) :: e1 →
is subtype (psubst env e2 x p ++ e1 )

(psubst type s x p) (psubst type t x p)).
induction 2; intros; simpl ; subst e.

Rule sa top
constructor. eapply wf env psubst ; eauto.
eapply wf type psubst ; eauto.

Rule sa re�
apply sub re�.
case (eq name x0 x ); intro.

Rule sa re�, case x = x0
apply wf type env incr with e1 ; eauto.
rewrite dom append. apply incl appr. apply incl re�.

Rule sa re�, case x 6= x0
constructor. auto. generalize (lookup inv H2 ).
repeat rewrite dom append. simpl. rewrite dom psubst env.
intro. apply in or app. elim (in app or H3 ); intro.
auto. elim H4 ; intro. congruence. auto.
eapply wf env psubst ; eauto.

Rule sa trans
assert (wf env (e2 ++ (x, q) :: e1 )). eapply is subtype wf env ; eauto.
generalize (lookup env concat p H3 H1 ). intro.
case (eq name x0 x ); intro.

Rule sa trans, case x = x0
subst x0. assert (u = q). intuition. subst u.
apply sub trans with (psubst type q x p).
replace (psubst type q x p) with q.
apply sub weaken with e1. auto. eapply wf env psubst ; eauto.
apply env concat weaken. eapply wf env psubst ; eauto.
symmetry. apply psubst type inv.
assert (wf type e1 q). eapply is subtype wf type r ; eauto.
generalize (fv wf type x H5 ).
generalize (env concat not free H3 ). tauto.
apply IHis subtype. auto.

Rule sa trans, case x 6= x0
apply sa trans tvar with (psubst type u x p).
auto. generalize (sym not equal n); tauto.
apply IHis subtype. auto.

Rule sa arrow
constructor ; auto.

Rule sa forall
destruct (fresh name TYPE (x :: dom (e2 ++ (x, q) :: e1 ) ++

dom (psubst env e2 x p ++ e1 ) ++
fv type (psubst type s2 x p) ++
fv type (psubst type t2 x p)))

as [y [F K]].
eapply sa all' ; eauto.
repeat rewrite (psubst freshen type e1 ).
change ((y, psubst type t1 x p) :: psubst env e2 x p ++ e1 )
with (psubst env ((y, t1 ) :: e2 ) x p ++ e1 ).

apply H2 ; eauto. eauto. eauto. eauto. eauto.

eauto.
Qed.

Chapter 3

Type soundness

This chapter addresses part 2A of the POPLmark challenge, namely the proof of soundness
of the F<: type systems without records.

Require Import Arith.
Require Import ZArith.
Require Import List.
Require Import extralibrary.
Require Import part1a.

3.1 Terms

We now de�ne the syntax of F<: terms and basic syntactic notions such as free variables,
substitutions, and well-formedness of terms. We follow the same approach used for types in
chapter 2.

3.1.1 Syntax and syntactic operations

The syntax of terms is de�ned as follows. As in types, bound variables are represented by
de Bruijn indices, while free variables are represented by names. Bound term variables and
bound type variables are numbered independently. In a lambda-abstraction TFun t a, the
term variable Var 0 is bound in a. In a type abstraction TApp t a, the type variable TVar
0 is bound in a.

Inductive term: Set :=
| Param: name → term
| Var : nat → term
| Fun: type → term → term

| App: term → term → term
| TFun: type → term → term
| TApp: term → type → term.

The free names of a term include both type names and term names.

Fixpoint fv term (a: term) : list name :=
match a with
| Param v ⇒ v :: nil
| Var n ⇒ nil
| Fun t a1 ⇒ fv type t ++ fv term a1
| App a1 a2 ⇒ fv term a1 ++ fv term a2
| TFun t a1 ⇒ fv type t ++ fv term a1
| TApp a1 t ⇒ fv term a1 ++ fv type t
end.

There are 4 substitution operations over terms, depending on whether we are substituting a
named variable (psubst ) or a de Bruijn variable (vsubst ), and whether we are substituting
a term for a term variable ( term) or a type for a type variable ( tety).

Fixpoint vsubst term (a: term) (x : nat) (b: term) {struct a} : term :=
match a with
| Param v ⇒ Param v
| Var n ⇒

match compare nat n x with
| Nat less ⇒ Var n
| Nat equal ⇒ b
| Nat greater ⇒ Var (pred n)
end

| Fun t a1 ⇒ Fun t (vsubst term a1 (S x ) b)
| App a1 a2 ⇒ App (vsubst term a1 x b) (vsubst term a2 x b)
| TFun t a1 ⇒ TFun t (vsubst term a1 x b)
| TApp a1 t ⇒ TApp (vsubst term a1 x b) t
end.

Fixpoint psubst term (a: term) (x : name) (b: term) {struct a} : term :=
match a with
| Param v ⇒ if eq name v x then b else Param v
| Var n ⇒ Var n
| Fun t a1 ⇒ Fun t (psubst term a1 x b)
| App a1 a2 ⇒ App (psubst term a1 x b) (psubst term a2 x b)
| TFun t a1 ⇒ TFun t (psubst term a1 x b)
| TApp a1 t ⇒ TApp (psubst term a1 x b) t
end.

Fixpoint vsubst tety (a: term) (x : nat) (b: type) {struct a} : term :=

match a with
| Param v ⇒ Param v
| Var n ⇒ Var n
| Fun t a1 ⇒ Fun (vsubst type t x b) (vsubst tety a1 x b)
| App a1 a2 ⇒ App (vsubst tety a1 x b) (vsubst tety a2 x b)
| TFun t a1 ⇒ TFun (vsubst type t x b) (vsubst tety a1 (S x ) b)
| TApp a1 t ⇒ TApp (vsubst tety a1 x b) (vsubst type t x b)
end.

Fixpoint psubst tety (a: term) (x : name) (b: type) {struct a} : term :=
match a with
| Param v ⇒ Param v
| Var n ⇒ Var n
| Fun t a1 ⇒ Fun (psubst type t x b) (psubst tety a1 x b)
| App a1 a2 ⇒ App (psubst tety a1 x b) (psubst tety a2 x b)
| TFun t a1 ⇒ TFun (psubst type t x b) (psubst tety a1 x b)
| TApp a1 t ⇒ TApp (psubst tety a1 x b) (psubst type t x b)
end.

Here are the two �freshening� operations that replace the bound variable 0 with a term or
type name, respectively.

De�nition freshen term (a: term) (x : name) : term :=
vsubst term a 0 (Param x ).

De�nition freshen tety (a: term) (x : name) : term :=
vsubst tety a 0 (Tparam x ).

Substitutions and freshening play well with free variables.

Lemma fv term vsubst term:
∀ x a n b, In x (fv term a) → In x (fv term (vsubst term a n b)).

Lemma fv term vsubst tety :
∀ x a n b, In x (fv term a) → In x (fv term (vsubst tety a n b)).

Lemma fv term freshen term:
∀ x a y, In x (fv term a) → In x (fv term (freshen term a y)).

Lemma fv term freshen tety :
∀ x a y, In x (fv term a) → In x (fv term (freshen tety a y)).

Swaps of two names in a term.

Fixpoint swap term (u v : name) (a: term) {struct a} : term :=
match a with
| Param x ⇒ Param (swap u v x )
| Var n ⇒ Var n

| Fun t a1 ⇒ Fun (swap type u v t) (swap term u v a1 )
| App a1 a2 ⇒ App (swap term u v a1 ) (swap term u v a2 )
| TFun t a1 ⇒ TFun (swap type u v t) (swap term u v a1 )
| TApp a1 t ⇒ TApp (swap term u v a1 ) (swap type u v t)
end.

Swaps commute with the free variables operation.

Lemma in fv term swap:
∀ u v x a,
In x (fv term a) ↔ In (swap u v x ) (fv term (swap term u v a)).

Lemma swap term not free:
∀ u v a, ¬In u (fv term a) → ¬In v (fv term a) → swap term u v a = a.

Swaps are self-inverse.

Lemma swap term inv : ∀ u v a, swap term u v (swap term u v a) = a.

Swaps commute with substitutions and freshening.

Lemma vsubst term swap:
∀ u v a n b,
swap term u v (vsubst term a n b) =
vsubst term (swap term u v a) n (swap term u v b).

Lemma vsubst tety swap:
∀ u v a n b,
swap term u v (vsubst tety a n b) =
vsubst tety (swap term u v a) n (swap type u v b).

Lemma freshen term swap:
∀ u v a x,
swap term u v (freshen term a x ) =
freshen term (swap term u v a) (swap u v x ).

Lemma freshen tety swap:
∀ u v a x,
swap term u v (freshen tety a x ) =
freshen tety (swap term u v a) (swap u v x ).

3.1.2 Well-formedness of terms

A term is well-formed in a typing environment if:

� all types contained within are well-formed as per wf type;

� all names n appearing free in a Param n subterm are of kind TERM and are bound
in the environment;

� it does not contain free de Bruijn variables.

Inductive wf term: typenv → term → Prop :=
| wf term param: ∀ e x,

kind x = TERM → In x (dom e) →
wf term e (Param x )

| wf term fun: ∀ e t a,
wf type e t →
(∀ x,
kind x = TERM →
¬In x (fv term a) → ¬In x (dom e) →
wf term ((x, t) :: e) (freshen term a x )) →
wf term e (Fun t a)

| wf term app: ∀ e a1 a2,
wf term e a1 → wf term e a2 →
wf term e (App a1 a2 )

| wf term tfun: ∀ e t a,
wf type e t →
(∀ x,
kind x = TYPE →
¬In x (fv term a) → ¬In x (dom e) →
wf term ((x, t) :: e) (freshen tety a x )) →
wf term e (TFun t a)

| wf term tapp: ∀ e a t,
wf term e a → wf type e t →
wf term e (TApp a t).

A term well formed in e has all its free names in the domain of e.

Lemma fv wf term: ∀ x e t, wf term e t → In x (fv term t) → In x (dom e).

Well-formedness is stable under swaps.

Lemma wf term swap:
∀ u v, kind u = kind v →
∀ e a, wf term e a → wf term (swap env u v e) (swap term u v a).

A term well-formed in e remains well-formed if extra bindings are added to e.

Lemma wf term env incr :
∀ e a, wf term e a → ∀ e', incl (dom e) (dom e' ) → wf term e' a.

Here are two admissible rules that prove the well-formedness of Fun and TFun abstractions.
These rules are similar to the wf term fun and wf term tfun rules, but with a premise of
the form �there exists a name� instead of the original �for all names�.

Lemma wf term fun' :
∀ e x t a,
wf type e t →
kind x = TERM → ¬In x (fv term a) → ¬In x (dom e) →
wf term ((x, t) :: e) (freshen term a x ) →
wf term e (Fun t a).

Lemma wf term tfun' :
∀ e x t a,
wf type e t →
kind x = TYPE → ¬In x (fv term a) → ¬In x (dom e) →
wf term ((x, t) :: e) (freshen tety a x ) →
wf term e (TFun t a).

3.1.3 Properties of term substitutions

To prove the usual properties of term substitutions, we follow the same approach as for type
substitutions, starting with a characterization of terms that have no free de Bruijn variables,
or all such variables below some threshold.

Fixpoint term vars below (a: term) (nterm ntype: nat) {struct a} : Prop :=
match a with
| Param x ⇒ True
| Var n ⇒ n < nterm
| Fun t b ⇒ type vars below t ntype ∧ term vars below b (S nterm) ntype
| App b c ⇒ term vars below b nterm ntype ∧ term vars below c nterm ntype
| TFun t b ⇒ type vars below t ntype ∧ term vars below b nterm (S ntype)
| TApp b t ⇒ term vars below b nterm ntype ∧ type vars below t ntype
end.

Lemma term vars below vsubst term:
∀ a nterm ntype a',
term vars below (vsubst term a nterm a' ) nterm ntype →
term vars below a (S nterm) ntype.

Lemma term vars below vsubst tety :
∀ a nterm ntype a',
term vars below (vsubst tety a ntype a' ) nterm ntype →
term vars below a nterm (S ntype).

Lemma wf term vars below 0 : ∀ e a, wf term e a → term vars below a 0 0.

Lemma vsubst term invariant below :
∀ a n1 n2 m b, term vars below a n1 n2 → n1 ≤ m → vsubst term a m b = a.

Lemma vsubst tety invariant below :
∀ a n1 n2 m t, term vars below a n1 n2 → n2 ≤ m → vsubst tety a m t = a.

Lemma vsubst term wf term:
∀ e a n b, wf term e a → vsubst term a n b = a.

Lemma vsubst tety wf term:
∀ e a n t, wf term e a → vsubst tety a n t = a.

Lemma psubst vsubst term:
∀ e a x b n c,
wf term e b →
vsubst term (psubst term a x b) n (psubst term c x b) =
psubst term (vsubst term a n c) x b.

Lemma psubst freshen term:
∀ e a x b y,
wf term e b → x 6= y →
freshen term (psubst term a x b) y = psubst term (freshen term a y) x b.

Lemma psubst vsubst tety :
∀ e a x b n c,
wf type e b →
vsubst tety (psubst tety a x b) n (psubst type c x b) =
psubst tety (vsubst tety a n c) x b.

Lemma psubst freshen tety :
∀ e a x b y,
wf type e b → x 6= y →
freshen tety (psubst tety a x b) y = psubst tety (freshen tety a y) x b.

Lemma psubst vsubst tetety :
∀ e a x b n c,
wf type e b →
vsubst term (psubst tety a x b) n (psubst tety c x b) =
psubst tety (vsubst term a n c) x b.

Lemma psubst freshen tetety :
∀ e a x b y,
wf type e b → x 6= y →
freshen term (psubst tety a x b) y = psubst tety (freshen term a y) x b.

Lemma psubst vsubst tetyte:
∀ e a x b n c,
wf term e b →
vsubst tety (psubst term a x b) n c = psubst term (vsubst tety a n c) x b.

Lemma psubst freshen tetyte:

∀ e a x b y,
wf term e b → x 6= y →
freshen tety (psubst term a x b) y = psubst term (freshen tety a y) x b.

Lemma vsubst psubst term:
∀ x a2 a1 n,
¬In x (fv term a1 ) →
vsubst term a1 n a2 = psubst term (vsubst term a1 n (Param x )) x a2.

Lemma vsubst psubst freshen term:
∀ x a1 a2,
¬In x (fv term a1 ) →
vsubst term a1 0 a2 = psubst term (freshen term a1 x ) x a2.

Lemma vsubst psubst tety :
∀ x t2 a1 n,
¬In x (fv term a1 ) →
vsubst tety a1 n t2 = psubst tety (vsubst tety a1 n (Tparam x )) x t2.

Lemma vsubst psubst freshen tety :
∀ x a1 t2,
¬In x (fv term a1 ) →
vsubst tety a1 0 t2 = psubst tety (freshen tety a1 x ) x t2.

3.2 Typing rules

We now de�ne the typing judgement �term a has type t in environment e� as an inductive
predicate has type e a t.

Inductive has type: typenv → term → type → Prop :=
| t var : ∀ e x t,

wf env e → kind x = TERM → lookup x e = Some t →
has type e (Param x ) t

| t abs: ∀ e t1 a t2,
wf type e t1 →
(∀ x,
kind x = TERM → ¬In x (dom e) →
has type ((x, t1 ) :: e) (freshen term a x ) t2 ) →

has type e (Fun t1 a) (Arrow t1 t2 )
| t app: ∀ e a b t1 t2,

has type e a (Arrow t1 t2 ) → has type e b t1 →
has type e (App a b) t2

| t tabs: ∀ e t1 a t2,
wf type e t1 →

(∀ x,
kind x = TYPE → ¬In x (dom e) →
has type ((x, t1 ) :: e) (freshen tety a x ) (freshen type t2 x )) →

has type e (TFun t1 a) (Forall t1 t2 )
| t tapp: ∀ e a t t1 t2,

has type e a (Forall t1 t2 ) →
is subtype e t t1 →
has type e (TApp a t) (vsubst type t2 O t)

| t sub: ∀ e a t1 t2,
has type e a t1 → is subtype e t1 t2 →
has type e a t2.

Well-formedness properties: if has type e a t holds, then e is a well-formed environment, t
a well-formed type and a a well-formed term.

Lemma has type wf env : ∀ e a t, has type e a t → wf env e.

Lemma wf type strengthen:
∀ e t, wf type e t →
∀ e', (∀ x, kind x = TYPE → In x (dom e) → In x (dom e' )) → wf type e' t.

Lemma has type wf type: ∀ e a t, has type e a t → wf type e t.

Lemma has type wf term: ∀ e a t, has type e a t → wf term e a.

The has type predicate is stable by addition of hypotheses.

Lemma wf type weaken: ∀ e e' t, wf type e t → env weaken e e' → wf type e' t.

Lemma wf term weaken: ∀ e e' a, wf term e a → env weaken e e' → wf term e' a.

Lemma env weaken add :
∀ e e' x t, env weaken e e' → env weaken ((x, t) :: e) ((x, t) :: e' ).

Lemma has type weaken:
∀ e a t, has type e a t → ∀ e', wf env e' → env weaken e e' → has type e' a t.

The has type predicate is equivariant, i.e. stable by swapping.

Lemma has type swap:
∀ u v, kind u = kind v →
∀ e a t, has type e a t →
has type (swap env u v e) (swap term u v a) (swap type u v t).

As a consequence of equivariance, we obtain admissible typing rules for functions and type
abstractions, similar to rules t abs and t tabs but where the variable name is quanti�ed
existentially rather than universally.

Lemma kind fv type: ∀ e t, wf type e t → ∀ x, In x (fv type t) → kind x = TYPE.

Lemma fv wf type kind : ∀ x e t, wf type e t → kind x = TERM → ¬In x (fv type t).

Lemma fresh freshen term:
∀ x t1 e a y,
wf term ((x, t1 ) :: e) (freshen term a x ) → ¬In y (dom e) → x 6= y →
¬In y (fv term a).

Lemma t abs' :
∀ e t1 a t2 x,
kind x = TERM → ¬In x (dom e) → ¬In x (fv term a) →
has type ((x, t1 ) :: e) (freshen term a x ) t2 →
has type e (Fun t1 a) (Arrow t1 t2 ).

Lemma fresh freshen tety :
∀ x t1 e a y,
wf term ((x, t1 ) :: e) (freshen tety a x ) → ¬In y (dom e) → x 6= y →
¬In y (fv term a).

Lemma t tabs' :
∀ e t1 a t2 x,
kind x = TYPE → ¬In x (dom e) → ¬In x (fv term a) → ¬In x (fv type t2 ) →
has type ((x, t1 ) :: e) (freshen tety a x ) (freshen type t2 x ) →
has type e (TFun t1 a) (Forall t1 t2 ).

3.3 Stability of the typing judgement under substitutions

We now show that the typing judgement is stable under substitutions. There are two
substitutions to consider: of a type for a type variable, and of a term for a term variable.

Lemma has type stable type subst :
∀ e1 x p q e2 a t,
kind x = TYPE →
is subtype e1 p q →
has type (e2 ++ (x, q) :: e1 ) a t →
has type (psubst env e2 x p ++ e1 ) (psubst tety a x p) (psubst type t x p).

Lemma lookup env append :
∀ e2 x p y e1,
wf env (e1 ++ (x, p) :: e2 ) →
lookup y (e1 ++ (x, p) :: e2 ) = if eq name y x then Some p else lookup y (e1 ++ e2 ).

Lemma wf env append :
∀ e2 x p e1, wf env (e1 ++ (x, p) :: e2 ) → kind x = TERM → wf env (e1 ++ e2 ).

Lemma is subtype strengthen:

∀ e s t, is subtype e s t →
∀ e', wf env e' → (∀ x : name, kind x = TYPE → lookup x e' = lookup x e) →
is subtype e' s t.

Lemma has type stable term subst :
∀ e1 x b s e2 a t,
kind x = TERM → has type e1 b s → has type (e2 ++ (x, s) :: e1 ) a t →
has type (e2 ++ e1 ) (psubst term a x b) t.

3.4 Dynamic semantics

The dynamic semantics of F<: is speci�ed by a one-step reduction relation, in small-step
operational style. We �rst de�ne values (�nal results of reduction sequences) as a subset of
terms.

Inductive isvalue: term → Prop :=
| isvalue fun: ∀ t a,

isvalue (Fun t a)
| isvalue tfun: ∀ t a,

isvalue (TFun t a).

We �rst give a Plotkin-style speci�cation of the reduction relation: it uses inductive rules
red appfun, red apparg, red tapp instead of contexts to describe reductions inside applica-
tions. The two rules red appabs and red tapptabs are the familiar beta-reduction rules for
term and type applications, respectively.

Inductive red : term → term → Prop :=
| red appabs: ∀ t a v,

isvalue v →
red (App (Fun t a) v) (vsubst term a 0 v)

| red tapptabs: ∀ t a t',
red (TApp (TFun t a) t' ) (vsubst tety a 0 t' )

| red appfun: ∀ a a' b,
red a a' → red (App a b) (App a' b)

| red apparg : ∀ v b b',
isvalue v → red b b' → red (App v b) (App v b' )

| red tapp: ∀ a a' t,
red a a' → red (TApp a t) (TApp a' t).

We now give an alternate speci�cation of the reduction relation in the style of Wright and
Felleisen. The red top relation captures beta-reductions at the top of a term. Reductions
within terms are expressed using reduction contexts (see the red context relation). Contexts
are represented as functions from terms to terms whose shape is constrained by the is context
predicate.

Inductive red top: term → term → Prop :=
| red top appabs: ∀ t a v,

isvalue v →
red top (App (Fun t a) v) (vsubst term a 0 v)

| red top tapptabs: ∀ t a t',
red top (TApp (TFun t a) t' ) (vsubst tety a 0 t' ).

Inductive is context : (term → term) → Prop :=
| iscontext hole:

is context (fun a ⇒ a)
| iscontext app left : ∀ c b,

is context c → is context (fun x ⇒ App (c x ) b)
| iscontext app right : ∀ v c,

isvalue v → is context c → is context (fun x ⇒ App v (c x ))
| iscontext tapp: ∀ c t,

is context c → is context (fun x ⇒ TApp (c x ) t).

Inductive red context : term → term → Prop :=
| red context intro: ∀ a a' c,

red top a a' → is context c → red context (c a) (c a' ).

The Plotkin-style relation is more convenient for doing formal proofs. Since the challenge
is given in terms of contexts, we feel obliged to prove the equivalence between the two
formulations of reduction. The proofs are routine inductions over the derivations of red and
is context, respectively.

Lemma red red context : ∀ a a', red a a' → red context a a'.

Lemma red context red : ∀ a a', red context a a' → red a a'.

3.5 Type soundness proof

Type soundness for F<: is established by proving the standard properties of type preservation
(also called subject reduction) and progress.

3.5.1 Preservation

Technical inversion lemmas on typing derivations. These lemmas are similar (but not fully
identical) to lemma A.13 in the on-paper proof.

Lemma has type fun inv :
∀ e a t, has type e a t →
∀ b s1 u1 u2, a = Fun s1 b → is subtype e t (Arrow u1 u2 ) →

is subtype e u1 s1 ∧
∃ s2,
is subtype e s2 u2 ∧
(∀ x, kind x = TERM → ¬In x (dom e) → has type ((x, s1 ) :: e) (freshen term b x )

s2 ).

Lemma has type tfun inv :
∀ e a t, has type e a t →
∀ b s1 u1 u2, a = TFun s1 b → is subtype e t (Forall u1 u2 ) →
is subtype e u1 s1 ∧
∃ s2,
(∀ x, kind x = TYPE → ¬In x (dom e) →

is subtype ((x, u1 ) :: e) (freshen type s2 x ) (freshen type u2 x )) ∧
(∀ x, kind x = TYPE → ¬In x (dom e) →

has type ((x, s1 ) :: e) (freshen tety b x ) (freshen type s2 x )).

The preservation theorem states that if term a reduces to a', then all typings valid for a are
also valid for a'. It is proved by an outer induction on the reduction and an inner induction
on the typing derivation (to get rid of subtyping steps).

Theorem preservation: ∀ e a a' t, red a a' → has type e a t → has type e a' t.
Proof.
assert (∀ a a', red a a' →

∀ e a0 t, has type e a0 t → ∀ (EQ : a = a0 ),
has type e a' t).

induction 1; induction 1; intros; simplify eq EQ ; clear EQ ; intros; subst ;
try (eapply t sub; eauto; fail).

Case app abs
assert (is subtype e (Arrow t1 t2 ) (Arrow t1 t2 )). apply sub re� ; eauto.
destruct (has type fun inv H0 (re� equal ) H0 )
as [A [s2 [B C]]].
apply t sub with s2 ; auto.
destruct (fresh name TERM (dom e ++ fv term a)) as [x [F K]].
rewrite (vsubst psubst freshen term x ); eauto.
change e with (nil ++ e).
apply has type stable term subst with t ; auto.
apply t sub with t1 ; auto.
simpl ; eauto.

Case tapp tabs
assert (is subtype e (Forall t1 t2 ) (Forall t1 t2 )). apply sub re� ; eauto.
destruct (has type tfun inv H (re� equal ) H1 )
as [A [s2 [B C]]].
destruct (fresh name TYPE (dom e ++ fv term a ++ fv type t2 )) as [x [F K]].
rewrite (vsubst psubst freshen tety x ); eauto.

rewrite (vsubst psubst freshen type x ); eauto.
apply t sub with (psubst type (freshen type s2 x ) x t0 ).
change e with (psubst env nil x t0 ++ e).
apply has type stable type subst with t ; eauto.
apply sub trans with t1 ; auto.
simpl ; auto.
change e with (psubst env nil x t0 ++ e).
apply sub stable subst with t1 ; eauto. simpl ; auto.

Case context left app
apply t app with t1 ; eauto.

Case context right app
apply t app with t1 ; eauto.

Case context left tapp
apply t tapp with t1 ; eauto.

Final conclusion
eauto.

Qed.

3.5.2 Progress

The following lemma, which corresponds to lemma A.14 in the challenge statement, deter-
mines the shape of a value from its type. Namely, closed values of function types are function
abstractions, and closed values of polymorphic types are type abstractions.

Lemma canonical forms:
∀ e a t, has type e a t → e = nil → isvalue a →
match t with
| Arrow t1 t2 ⇒ ∃ s, ∃ b, a = Fun s b
| Forall t1 t2 ⇒ ∃ s, ∃ b, a = TFun s b
| Top ⇒ True
| ⇒ False
end.

The progress theorem shows that a term well-typed in the empty environment is never
�stuck�: either it is a value, or it can reduce. The theorem is proved by a simple induction
on the typing derivation for the term and a case analysis on whether the subterms of the
term are values or can reduce further.

Theorem progress: ∀ a t, has type nil a t → isvalue a ∨ ∃ a', red a a'.
Proof.
assert (∀ e a t, has type e a t → e = nil → isvalue a ∨ exists a', red a a' ).
induction 1; intros; subst e.

Free variable: impossible in the empty typing environment.

simpl in H1. discriminate.
Function: already a value.
left ; constructor.

Application App a b.
right.
destruct (IHhas type1 (re� equal )) as [Va | [a' Ra]].
destruct (IHhas type2 (re� equal )) as [Vb | [b' Rb]].

a and b are values. a must be a Fun. Beta-reduction is possible.
generalize (canonical forms H (re� equal ) Va).
intros [s [c EQ]]. subst a.
exists (vsubst term c 0 b). constructor. auto.

a is a value, but b reduces. App a b therefore reduces.
exists (App a b' ). constructor ; auto.

a reduces. App a b reduces as well.
exists (App a' b). constructor ; auto.

Type abstraction: already a value.
left ; constructor.

Type application TApp a t.
right. destruct (IHhas type (re� equal )) as [Va | [a' Ra]].

a is a value. a must be a TFun. Beta-reduction is possible.
generalize (canonical forms H (re� equal ) Va).
intros [s [b EQ]]. subst a.
exists (vsubst tety b 0 t). constructor.

a reduces, and so does TApp a t.
exists (TApp a' t). constructor ; auto.

Subtyping step.
auto.

Final conclusion.
eauto.

Qed.

Chapter 4

Execution of the dynamic

semantics

In this chapter, we consider the problem of executing F<: terms as prescribed by the reduc-
tion semantics for this language. Such executions are useful for testing that the semantics
has the intended behavior. This goal is listed as part 3 in the POPLmark challenge. As we
will see, our development will go one step further and result in the production of an e�cient
and provably correct interpreter for F<:.

There are two approaches to executing dynamic semantics within Coq. The �rst operates
directly on a relational speci�cation of the semantics, either big-step or small-step like our
red predicate from chapter 3. The eauto Coq tactic, which build proofs by Prolog-style
resolution over a set of predeclared inference rules and lemmas, can be abused to search and
build derivation trees for a goal of the form ∃ b, red a b, therefore executing one reduction
step from a. An example of this approach can be found in our work with A. Appel on
the list-machine benchmark [AL06]. However, this approach is tricky to set up and very
ine�cient.

The other approach, which we follow in this chapter, is to specify the operational seman-
tics as functions rather than predicates. While Coq has no e�cient built-in execution mech-
anism for logic programs (composed of inductively-de�ned predicates), it can natively evalu-
ate functional programs (composed of functions de�ned by recursion and pattern-matching).
Such functional reductions are actually part of the logic of Coq, via the notion of conversion.

We therefore proceed in two steps. We will �rst de�ne functions that compute the one-
step or N -step reduct of a F<: term, and prove that they are correct and complete with
respect to the relational semantics. We will then use these functions to evaluate terms within
Coq and to extract e�cient Caml code for an interpreter.

Require Import Arith.
Require Import ZArith.

Require Import List.
Require Import extralibrary.
Require Import part1a.
Require Import part2a.

4.1 Execution of one-step reductions

We �rst show that the isvalue predicate is decidable. The lemma below will actually provides
us with a decision procedure that takes any term a and returns whether it is a value or not.
We can then use this decision procedure within function de�nitions.

Lemma isvalue dec:
∀ a, {isvalue a} + {�isvalue a}.

The reduce function maps a term a to either Some b if a reduces in one step to b, or to
None if a does not reduce. It is de�ned by structural recursion over a and case analysis on
whether subterms of a are values, or reduce, or are stuck.

Fixpoint reduce (a: term) : option term :=
match a with
| App b c ⇒

if isvalue dec b then
if isvalue dec c then
match b with Fun t d ⇒ Some (vsubst term d 0 c) | ⇒ None end

else
match reduce c with Some c' ⇒ Some(App b c' ) | None ⇒ None end

else
match reduce b with Some b' ⇒ Some(App b' c) | None ⇒ None end

| TApp b t ⇒
if isvalue dec b then
match b with TFun t' c ⇒ Some (vsubst tety c 0 t) | ⇒ None end

else
match reduce b with Some b' ⇒ Some(TApp b' t) | None ⇒ None end

| ⇒ None
end.

We then show that this function is correct and complete with respect to the reduction rules:
reduce a = Some b if and only if red a b holds. The proofs are routine inductions on the
structure of a for the �only if� part and on the derivation of red a b for the �if� part.

Lemma reduce is correct :
∀ a a', reduce a = Some a' → red a a'.

Lemma isvalue dec true:

∀ a (T : Set) (b c: T ), isvalue a → (if isvalue dec a then b else c) = b.

Lemma isvalue dec false:
∀ a a' (T : Set) (b c: T ), red a a' → (if isvalue dec a then b else c) = c.

Lemma reduce is complete:
∀ a a', red a a' → reduce a = Some a'.

4.2 Execution of N -step reductions

The following function iterates the one-step reduction function compute to obtain the normal
form of a term. Since Coq functions must always terminate, we need to bound the number
of iterations by the n parameter. If a normal form cannot be reached in n steps, compute
returns None.

Fixpoint compute (n: nat) (a: term) {struct n}: option term :=
match n with
| O ⇒ None
| S n' ⇒

match reduce a with
| Some a' ⇒ compute n' a'
| None ⇒ Some a
end

end.

We now show that compute a, if it succeeds, returns a reduct of a that is in normal form
(irreducible).

De�nition irreducible (a: term): Prop := ∀ b, ¬red a b.

Inductive red sequence: term → term → Prop :=
| red sequence 0 :
∀ a, irreducible a → red sequence a a

| red sequence 1 : ∀ a b c,
red a b → red sequence b c → red sequence a c.

Lemma compute correct :
∀ n a a', compute n a = Some a' → red sequence a a'.

Conversely, if a term a has a normal form a', there exists a number of iterations n such that
compute returns Some a'.

Lemma compute complete:
∀ a a', red sequence a a' → ∃ n, compute n a = Some a'.

4.3 Experiments

We can now use the Coq directives Eval compute in (reduce a) and Eval compute in (compute
N a) to display the results of performing one or N reduction steps in a.

De�nition F poly identity := TFun Top (Fun (Tvar 0) (Var 0)).
De�nition F top identity := TApp F poly identity Top.
De�nition F delta := Fun (Arrow Top Top) (App (Var 0) (Var 0)).
De�nition F testprog := App F delta F top identity.

Eval compute in (reduce F testprog).
Eval compute in (compute 100 F testprog).

The latter returns Some (Fun Top (Var 0)), which is indeed the value of the term F testprog.
For a larger example, here is some arithmetic on Church integers.

De�nition F one : term :=
(TFun Top (TFun (Tvar 0) (TFun (Tvar 1)

(Fun (Arrow (Tvar 2) (Tvar 1))
(Fun (Tvar 0)
(App (Var 1) (Var 0))))))).

De�nition F nat : type :=
(Forall Top
(Forall (Tvar 0)
(Forall (Tvar 1)
(Arrow (Arrow (Tvar 2) (Tvar 1)) (Arrow (Tvar 0) (Tvar 1)))))).

De�nition F add : term :=
(Fun F nat
(Fun F nat
(TFun Top (TFun (Tvar 0) (TFun (Tvar 1)

(Fun (Arrow (Tvar 2) (Tvar 1))
(Fun (Tvar 0)

(App (TApp (TApp (TApp (Var 3) (Tvar 2)) (Tvar 1)) (Tvar 1))
(App (Var 1)

(App (TApp (TApp (TApp (Var 2) (Tvar 2)) (Tvar 1))
(Tvar 0))

(App (Var 1) (Var 0)))))))))))).

Eval compute in (compute 100 (App (App F add F one) F one)).

Execution is nearly instantaneous. In Coq 8.1, we can also use Eval vm compute to re-
quest evaluation via compilation to virtual machine code. This results in execution speed
comparable to that of bytecoded OCaml.

An alternate execution path is to generate (or �extract� in Coq's terminology) Caml code
from the Coq de�nition of function compute. This is achieved by the following command:

Extraction "/tmp/fsub eval.ml" compute.

The generated Caml code can be compiled with the OCaml native-code compiler for even
higher execution speed. More importantly, it can be linked with a lexer, parser and printer
hand-written in OCaml, obtaining a stand-alone reference interpreter for F<: that can exe-
cute non-trivial programs.

Chapter 5

Assessment

The POPLmark challenge is still ongoing, and the many solutions submitted di�er widely
� in the encodings of binders used, in the proof assistants used, and even in the proof styles
of each author. It is therefore too early to draw conclusions. In this �nal chapter, we try to
assess the quality of our solution with respect to various criteria.

Legibility of de�nitions and theorems Mechanized proofs often de�ne notions and
state theorems in ways that are somewhat di�erent from what a mathematician would do
in an on-paper proof. Often, the mechanized de�nitions and statements are more precise,
but also harder to read and to relate with one's intuitions. The locally nameless approach
followed in this report remains quite close to the on-paper de�nitions and theorems given
in the statement of the POPLmark challenge. De�nitions and statements are mostly un-
surprising, except perhaps for the cases of de�nitions that involve crossing a binder. For
instance, de�nition of subtyping between ∀ types is written on paper as

Γ ` τ1 <: σ1 Γ, X <: τ1 ` σ2 <: τ2

Γ ` (∀X <: σ1. σ2) <: (∀X <: τ1. τ2)

while in our approach it is written as

Γ ` τ1 <: σ1 ∀X, X /∈ Dom(Γ) =⇒ Γ, X <: τ1 ` σ2[0← X] <: τ2[0← X]

Γ ` (∀ <: σ1. σ2) <: (∀ <: τ1. τ2)

There are several �tricks� here that surprise the reader: the de Bruijn notation for the
bound variable, the substitutions [0← X] in the second premise, and the placement of the
quanti�cation over X. In contrast, a purely nominal approach such as the solution by Urban
et al. leads to a less mysterious de�nition of the form

Γ ` τ1 <: σ1 X fresh for Γ Γ, X <: τ1 ` σ2 <: τ2

Γ ` (∀X <: σ1. σ2) <: (∀X <: τ1. τ2)

Nonetheless, we believe that de�nitions and theorems written in the locally nameless style
are globally more readable and intuitively understandable than their equivalents using either
de Bruijn indices or higher-order abstract syntax.

Overheads Mechanized proofs must make explicit a great number of small details and
obvious properties that are omitted in research papers. In the case of type systems and
operational semantics, this includes basic properties of terms, binders (α-conversion), sub-
stitutions, typing environments, and reduction sequences. All these properties had to be
painfully made explicit in the present development. We estimate that such sca�olding work
represent more than half of our development. More precisely, we consider that the following
parts are pure overhead, compared with a paper proof:

� De�nition and properties of free variables.

� De�nition and properties of substitutions. There are two substitutions per data type
of interest (one for free names and the other for bound de Bruijn variables), with the
associated commutation properties.

� De�nition and properties of swaps.

� Proving equivariance for many de�nitions.

� Proving admissibility of the rules with ∃X . . . in the premises, instead of ∀X . . .

� Manipulations of environments, e.g. properties of environments of the form Γ1, X <:
τ, Γ2.

A legitimate question to ask is: how much of this overhead could be avoided, either by
factoring the de�nitions and properties in reusable libraries, or by generating them auto-
matically from high-level descriptions of the syntax of terms? Examples of reusable libraries
include Urban and Tasson's nominal package, based on Isabelle/HOL type classes [UT05],
and Chlipala's library for typing environments. However, such forms of reuse are limited
in Coq by lack of type classes and any other form of �polytypic� de�nitions. Automatic
generation in the style of Pottier's Cαml [Pot06] is another direction that remains to be
investigated.

Size of the development Our solution is neither particularly compact nor excessively
verbose compared with other solutions. Arthur Charguéraud compared the sizes of several
Coq solutions to part 1A of the challenge, counting the number of non-trivial tactics invoked.
The results are as follows:

Authors Tactics Representation used
Jérome Vouillon 431 de Bruijn indices
Aaron Stump 1147 names and levels
Xavier Leroy 630 locally nameless
Hirschowitz & Maggesi 1615 de Bruijn indices
Adam Chlipala 342 locally nameless
Arthur Charguéraud 233 locally nameless

Several factors contribute to this relative verbosity of our solution. First, as we discussed
above, the locally nameless approach comes with inherent overheads. Second, as pointed out
by Charguéraud, there are small variations of the locally nameless approach that can reduce
the proof e�ort signi�cantly (see below). Finally, unlike Vouillon, Chlipala and Charguéraud,
we did not really take advantage of Coq's facilities for de�ning domain-speci�c tactics.

Possible improvements to the locally nameless approach Charguéraud [Cha06] and
Charguéraud, Pierce and Weirich [CPW06] present several variations on the locally nameless
approach that could signi�cantly simplify our development. The most important one is to
de�ne predicates in the binder-crossing by quanti�cation over all names not in a given �nite
set L of names, rather than all fresh names:

∀L,
Γ ` τ1 <: σ1 ∀X, X /∈ L =⇒ Γ, X <: τ1 ` σ2[0← X] <: τ2[0← X]

Γ ` (∀ <: σ1. σ2) <: (∀ <: τ1. τ2)

With this simple device, it is no longer necessary to prove equivariance of the well-formedness
and subtyping relations. In turn, this removes the need to de�ne and reason about swaps
of names. The downside of this approach is reduced legibility: it becomes even less obvious
why this rule captures the correct notion of subtyping between quanti�ed types.

The second simpli�cation is to treat well-formed typing environments as unordered sets
of bindings, rather than ordered lists. For instance, instead of reasoning over environments
of the form Γ1, Γ2, it is easier and often su�cient to reason over environments Γ that contain
all bindings from Γ1 and all bindings from Γ2. Similarly, well-formedness of terms and types
can pro�tably be de�ned with respect to a set of names bound in an environment rather
than with respect to an environment.

Finally, the de�nition of the vsubst substitutions (replacement of a bound variable by
a term) can also be simpli�ed: instead of taking

vsubst type (Tvar n) x b =


Tvar n, if n < x
b, if n = x
Tvar (n− 1), if n > x

it su�ces to take

vsubst type (Tvar n) x b =
{
Tvar n, if n 6= x
b, if n = x

since the case n > x is never exercised in the rest of the development. This simple change
signi�cantly simpli�es the proofs of the commutation lemmas between psubst and vsubst

substitutions.

Executability of the semantics Part 3 of the POPLmark challenge (testing and animat-
ing the dynamic semantics) did not receive much attention from the participants. However,
we believe that it is generally important to generate correct-by-construction implementa-
tions of programming tools from high-level speci�cations. The Coq proof assistant takes
an almost schizophrenic stance on this issue. On the one hand, Coq provides excellent ex-
ecution facilities for de�nitions written in functional style, both via its built-in evaluator
(using an e�cient virtual machine since version 8.1) and via its code extraction facility.
In other work [Ler06, BDL06], we developed a whole optimizing C compiler by extraction
from Coq functional speci�cations. On the other hand, Coq supports poorly the execution
of speci�cations written in relational style (inductive predicates), forcing users to manually
write functional variants of their relational speci�cations and to prove the equivalence of
the functional and relational speci�cations, as we did in chapter 4. This is an annoyance in
general, but also an interesting feature in some cases. In particular, it encourages a style
where relational speci�cations are not polluted by executability constraints, and can e.g.
be non-deterministic or even undecidable; executability constraints can then be taken into
account later, in typical re�nement style, when developing the functional speci�cation.

Extension to records and record types The POPLmark challenge suggests to extend
system F<: with records and record types with depth and width subtyping (parts 1B and 2B
of the challenge). We considered brie�y this extension but did not pursue it. This extension
is conceptually rather easy but technically di�cult: as in the other solutions that addressed
parts 1B and 2B of the challenge, it appears necessary to de�ne types and record types, as
well as terms and record terms, in a mutually recursive manner. Such mutually recursive
de�nitions are quite painful to handle in Coq, and require extensive changes to the proof
scripts. Again, this is more a limitation of Coq than a fundamental di�culty.

